

Esercizio

Si consideri un punto P di massa m mobile lungo una guida rettilinea liscia verticale e sia x la coordinata di P lungo un asse parallelo alla guida ed orientato verso il basso. Sul punto, oltre al peso ed alla reazione vincolare, agiscono una forza di attrito di tipo viscoso (proporzionale alla velocità) ed una forza elastica non lineare. L'equazione del moto di P si ottiene proiettando lungo la guida l'equazione di Newton e risulta:

$$m \ddot{x} = m g - \alpha \dot{x} + f_{el}(x) \quad (1)$$

dove g è l'accelerazione di gravità, α il coefficiente (positivo) di attrito viscoso e

$$f_{el}(x) = -k(x + cx^3) \quad k, c > 0$$

la forza elastica.

- (1) Dimostrare che il sistema ha una sola configurazione di equilibrio, x_{eq} , e che x_{eq} può essere approssimato con il metodo di bisezione.
- (2) Posto $g = 9.81 \text{ m/s}^2$, $m = 1 \text{ kg}$, $k = 10 \text{ N/m}$, $c = 10^{-2} \text{ m}^{-2}$ e $\alpha = 0.7 \text{ Ns/m}$, utilizzare la procedura definita in `bisezione.sci`¹ per determinare un'approssimazione di x_{eq} con errore assoluto inferiore a 10^{-5} m .
- (3) Posto $z = x - x_{eq}$, scrivere l'equazione del moto (1) in termini di z .
- (4) Nell'equazione (differenziale non lineare) scritta al punto precedente, compare l'addendo $f_{el}(x_{eq} + z)$. Sostituire tale termine con il polinomio di ordine uno ottenuto dallo sviluppo di Taylor di $f_{el}(x)$ in $x = x_{eq}$:

$$p(z) = f_{el}(x_{eq}) + f'_{el}(x_{eq}) z$$

e scrivere la corrispondente equazione differenziale (lineare ed omogenea). Le soluzioni di quest'ultima approssimano i moti del punto "per piccoli scostamenti dal moto di equilibrio $x(t) = x_{eq}$."

- (5) Si consideri l'equazione differenziale lineare ottenuta nel punto precedente. La soluzione dell'equazione che verifica le condizioni iniziali $z(0) = 0.1 \text{ m}$ e $\dot{z}(0) = 0 \text{ m/s}$ può essere approssimata, sull'intervallo di tempo $[0, 5] \text{ s}$, con il metodo di Eulero esplicito. Utilizzare la procedura definita in `LMV_TS_1_pv.sci` imponendo un errore locale massimo consentito pari a 10^{-1} . Determinare il numero di passi N_1 effettuati dalla procedura ed utilizzare il comando `plot` per ottenere i grafici della posizione (z) e della velocità (\dot{z}) in funzione del tempo e del passo in funzione dell'istante di integrazione..

¹I files `.sci` a cui si fa riferimento si trovano sulla pagina web del corso nella sezione **altro materiale didattico**.

- (6) Determinare la soluzione esatta dell'equazione differenziale lineare e verificare che descrive il moto di un oscillatore armonico *smorzato*. Ci si aspetta che le funzioni posizione e velocità siano oscillazioni di ampiezza *decescente* nel tempo. Si constata invece che le soluzioni numeriche hanno andamento opposto (oscillazioni di ampiezza *crescente* nel tempo).
- (7) Ripetere l'integrazione numerica imponendo un errore locale massimo consentito pari a 10^{-i} con $i = 2, \dots, 5$. Per ciascuna delle integrazioni, determinare il numero di passi N_i che la procedura ha effettuato ed il valore massimo della norma dell'errore totale ET_i .
- (8) Calcolare i rapporti N_{i+2}/N_i e ET_{i+2}/ET_i per $i = 1, 2$ e 3 . Giustificare i risultati ottenuti ed indicare quali sono, approssimativamente, il numero di passi e la norma dell'errore totale massimo che si dovrebbero ottenere imponendo un errore locale massimo consentito pari a 10^{-9} .

Facoltativo: Utilizzare la procedura definita in `LMV_TS_1_pv.sci` per approssimare la soluzione del problema di Cauchy *non lineare* costituito dall'equazione differenziale ottenuta al punto (3) e dalle condizioni iniziali $z(0) = 0.1$ m e $\dot{z}(0) = 0$ m/s, imponendo un errore locale massimo consentito pari a 10^{-5} . Si confronti poi l'approssimazione ottenuta con quella relativa al problema *lineare* ottenuta al punto (7) con lo stesso errore locale massimo consentito. Infine, ripetere il confronto imponendo come condizioni iniziali: $z(0) = 10$ m e $\dot{z}(0) = 0$ m/s.

```

//
clear;
//
Percorso = "/home/ciampa/Lavoro/Scilab/Lavoro/LMV/";
//
// costanti
//
g = 9.81;
m = 1;
k = 10;
c = 1d-2;
alfa = 0.7;
//
// STATICA *****
//
printf('\nApprossimazione della configurazione di equilibrio.\n\n');
//
function y=G(x)
    y = g - (k/m)*(x + c*x.^3)
endfunction
//
exec(Percorso + 'Zeri/bisezione.sci');
//
a = 0;
b = m*g/k;
tol = 1d-5;
kmax = 60;
//
[xeq,v,info] = bisezione(G,a,b,"assoluto",tol,kmax,"zitto");
//
printf("xeq = ");disp(xeq);
printf("v = ");disp(v);
printf("info = ");disp(info);
//
// DINAMICA LINEARE *****
//
printf('\nDinamica lineare.\n\n');
//
B = -k*(1 + 3*c*xeq^2);
A = [ 0, 1;
     B/m, -alfa/m];
//
function y=d1x(t, x)
    y = A * x
endfunction
//
function y=d2x(t, x)
    y = A * d1x(t,x)
endfunction
//
exec(Percorso + 'ODE/LMV_TS_1_pv.sci');
//
// Condizioni iniziali, errore locale massimo
//
x0 = [0.1;0];
t0 = 0;
tf = 5;
EL_MAX = 1d-5;
//
[T,U,PASSO] = LMV_TS_1_pv(x0,t0,tf,d1x,d2x,EL_MAX,"zitto");
//
// Soluzione esatta
//
rad = spec(A);
si = real(rad(1));
om = abs(imag(rad(1)));
//
function y=x_e(t)
    y = [x0(1)*exp(si*t) .* (-(si/om)*sin(om*t)+cos(om*t));
        -x0(1)*((si^2+om^2)/om)*exp(si*t) .* sin(om*t)]
endfunction
//
XE = x_e(T);
et = U - XE;
ET(1) = 0;
for i=2:length(T),
    ET(i) = norm(et(:,i));
end;
printf("\nerrore totale massimo = %3.2e\n",max(ET));
printf("\nnumero passi = %5d\n",length(PASSO));

```

```

//
// grafici
//
scf(0); clf();
//
subplot(211);
plot(T,U(1,:), "r", T,U(2,:), "b");
legend("posizione LIN", "velocità LIN");
xgrid();
format('e'); Titolo = 'EL_MAX = ' + string(EL_MAX); format('v');
xlabel(Titolo)
//
subplot(212);
plot(T(1:$-1), PASSO, "b");
xgrid();
xlabel("passo LIN");
//
//
// DINAMICA NON LINEARE *****
//
function y=f_el(x)
    y = -k*(x + c*x^3)
endfunction
//
function y=d1x_nl(t, x)
    y = [x(2);
        -(alfa/m)*x(2) + f_el(xeq + x(1))/m + g]
endfunction
//
function y=d2x_nl(t, x)
    z = -(alfa/m)*x(2) + f_el(xeq + x(1))/m + g
    y = [z;
        -(alfa/m)*z + (-k + 3*c*(xeq + x(1))^2)*x(2)/m]
endfunction
//
[T_nl,U_nl,PASSO_nl] = LMV_TS_1_pv(x0,t0,tf,d1x_nl,d2x_nl,EL_MAX,"zitto");
//
// grafici
//
scf(1); clf();
//
subplot(211);
plot(T_nl,U_nl(1,:), "g", T,U(1,:), "r");
legend("posizione NL", "posizione LIN");
xgrid();
//
subplot(212);
plot(T_nl,U_nl(2,:), "g", T,U(2,:), "b");
legend("velocità NL", "velocità LIN");
xgrid();
xlabel(Titolo);
//

```